
NAG Library Routine Document

D03PCF/D03PCA

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of bold italicised terms
and other implementation-dependent details.

1 Purpose

D03PCF/D03PCA integrates a system of linear or nonlinear parabolic partial differential equations
(PDEs) in one space variable. The spatial discretization is performed using finite differences, and the
method of lines is employed to reduce the PDEs to a system of ordinary differential equations (ODEs).
The resulting system is solved using a backward differentiation formula method.

D03PCA is a version of D03PCF that has additional arguments in order to make it safe for use in
multithreaded applications (see Section 5).

2 Specification

2.1 Specification for D03PCF

SUBROUTINE D03PCF (NPDE, M, TS, TOUT, PDEDEF, BNDARY, U, NPTS, X, ACC,
RSAVE, LRSAVE, ISAVE, LISAVE, ITASK, ITRACE, IND,
IFAIL)

&
&

INTEGER NPDE, M, NPTS, LRSAVE, ISAVE(LISAVE), LISAVE, ITASK,
ITRACE, IND, IFAIL

&

REAL (KIND=nag_wp) TS, TOUT, U(NPDE,NPTS), X(NPTS), ACC, RSAVE(LRSAVE)
EXTERNAL PDEDEF, BNDARY

2.2 Specification for D03PCA

SUBROUTINE D03PCA (NPDE, M, TS, TOUT, PDEDEF, BNDARY, U, NPTS, X, ACC,
RSAVE, LRSAVE, ISAVE, LISAVE, ITASK, ITRACE, IND,
IUSER, RUSER, CWSAV, LWSAV, IWSAV, RWSAV, IFAIL)

&
&

INTEGER NPDE, M, NPTS, LRSAVE, ISAVE(LISAVE), LISAVE, ITASK,
ITRACE, IND, IUSER(*), IWSAV(505), IFAIL

&

REAL (KIND=nag_wp) TS, TOUT, U(NPDE,NPTS), X(NPTS), ACC,
RSAVE(LRSAVE), RUSER(*), RWSAV(1100)

&

LOGICAL LWSAV(100)
CHARACTER(80) CWSAV(10)
EXTERNAL PDEDEF, BNDARY

3 Description

D03PCF/D03PCA integrates the system of parabolic equations:

XNPDE
j¼1

Pi;j
@Uj

@t
þQi ¼ x�m @

@x
xmRið Þ; i ¼ 1; 2; . . . ;NPDE; a � x � b; t � t0; ð1Þ

where Pi;j, Qi and Ri depend on x, t, U , Ux and the vector U is the set of solution values

U x; tð Þ ¼ U1 x; tð Þ; . . . ; UNPDE x; tð Þ½ �T; ð2Þ
and the vector Ux is its partial derivative with respect to x. Note that Pi;j, Qi and Ri must not depend

on
@U

@t
.

The integration in time is from t0 to tout, over the space interval a � x � b, where a ¼ x1 and b ¼ xNPTS

are the leftmost and rightmost points of a user-defined mesh x1; x2; . . . ; xNPTS. The coordinate system in
space is defined by the value of m; m ¼ 0 for Cartesian coordinates, m ¼ 1 for cylindrical polar

coordinates and m ¼ 2 for spherical polar coordinates. The mesh should be chosen in accordance with
the expected behaviour of the solution.

The system is defined by the functions Pi;j, Qi and Ri which must be specified in PDEDEF.

The initial values of the functions U x; tð Þ must be given at t ¼ t0. The functions Ri, for
i ¼ 1; 2; . . . ;NPDE, which may be thought of as fluxes, are also used in the definition of the boundary
conditions for each equation. The boundary conditions must have the form

�i x; tð ÞRi x; t; U; Uxð Þ ¼ �i x; t; U; Uxð Þ; i ¼ 1; 2; . . . ;NPDE; ð3Þ
where x ¼ a or x ¼ b.

The boundary conditions must be specified in BNDARY.

The problem is subject to the following restrictions:

(i) t0 < tout, so that integration is in the forward direction;

(ii) Pi;j, Qi and the flux Ri must not depend on any time derivatives;

(iii) the evaluation of the functions Pi;j, Qi and Ri is done at the mid-points of the mesh intervals by
calling the PDEDEF for each mid-point in turn. Any discontinuities in these functions must
therefore be at one or more of the mesh points x1; x2; . . . ; xNPTS;

(iv) at least one of the functions Pi;j must be nonzero so that there is a time derivative present in the
problem; and

(v) if m > 0 and x1 ¼ 0:0, which is the left boundary point, then it must be ensured that the PDE
solution is bounded at this point. This can be done by either specifying the solution at x ¼ 0:0 or
by specifying a zero flux there, that is �i ¼ 1:0 and �i ¼ 0:0. See also Section 9.

The parabolic equations are approximated by a system of ODEs in time for the values of Ui at mesh
points. For simple problems in Cartesian coordinates, this system is obtained by replacing the space
derivatives by the usual central, three-point finite difference formula. However, for polar and spherical
problems, or problems with nonlinear coefficients, the space derivatives are replaced by a modified
three-point formula which maintains second-order accuracy. In total there are NPDE� NPTS ODEs in
the time direction. This system is then integrated forwards in time using a backward differentiation
formula method.

4 References

Berzins M (1990) Developments in the NAG Library software for parabolic equations Scientific
Software Systems (eds J C Mason and M G Cox) 59–72 Chapman and Hall

Berzins M, Dew P M and Furzeland R M (1989) Developing software for time-dependent problems
using the method of lines and differential-algebraic integrators Appl. Numer. Math. 5 375–397

Dew P M and Walsh J (1981) A set of library routines for solving parabolic equations in one space
variable ACM Trans. Math. Software 7 295–314

Skeel R D and Berzins M (1990) A method for the spatial discretization of parabolic equations in one
space variable SIAM J. Sci. Statist. Comput. 11(1) 1–32

5 Arguments

1: NPDE – INTEGER Input

On entry: the number of PDEs in the system to be solved.

Constraint: NPDE � 1.

2: M – INTEGER Input

On entry: the coordinate system used:

M ¼ 0
Indicates Cartesian coordinates.

M ¼ 1
Indicates cylindrical polar coordinates.

M ¼ 2
Indicates spherical polar coordinates.

Constraint: M ¼ 0, 1 or 2.

3: TS – REAL (KIND=nag_wp) Input/Output

On entry: the initial value of the independent variable t.

On exit: the value of t corresponding to the solution values in U. Normally TS ¼ TOUT.

Constraint: TS < TOUT.

4: TOUT – REAL (KIND=nag_wp) Input

On entry: the final value of t to which the integration is to be carried out.

5: PDEDEF – SUBROUTINE, supplied by the user. External Procedure

PDEDEF must compute the functions Pi;j, Qi and Ri which define the system of PDEs. PDEDEF
is called approximately midway between each pair of mesh points in turn by D03PCF/D03PCA.

The specification of PDEDEF for D03PCF is:

SUBROUTINE PDEDEF (NPDE, T, X, U, UX, P, Q, R, IRES)

INTEGER NPDE, IRES
REAL (KIND=nag_wp) T, X, U(NPDE), UX(NPDE), P(NPDE,NPDE),

Q(NPDE), R(NPDE)
&

The specification of PDEDEF for D03PCA is:

SUBROUTINE PDEDEF (NPDE, T, X, U, UX, P, Q, R, IRES, IUSER,
RUSER)

&

INTEGER NPDE, IRES, IUSER(*)
REAL (KIND=nag_wp) T, X, U(NPDE), UX(NPDE), P(NPDE,NPDE),

Q(NPDE), R(NPDE), RUSER(*)
&

1: NPDE – INTEGER Input

On entry: the number of PDEs in the system.

2: T – REAL (KIND=nag_wp) Input

On entry: the current value of the independent variable t.

3: X – REAL (KIND=nag_wp) Input

On entry: the current value of the space variable x.

4: UðNPDEÞ – REAL (KIND=nag_wp) array Input

On entry: UðiÞ contains the value of the component Ui x; tð Þ, for i ¼ 1; 2; . . . ;NPDE.

5: UXðNPDEÞ – REAL (KIND=nag_wp) array Input

On entry: UXðiÞ contains the value of the component
@Ui x; tð Þ

@x
, for i ¼ 1; 2; . . . ;NPDE.

6: PðNPDE;NPDEÞ – REAL (KIND=nag_wp) array Output

On exit: Pði; jÞ must be set to the value of Pi;j x; t; U; Uxð Þ, for i ¼ 1; 2; . . . ;NPDE and
j ¼ 1; 2; . . . ;NPDE.

7: QðNPDEÞ – REAL (KIND=nag_wp) array Output

On exit: QðiÞ must be set to the value of Qi x; t; U; Uxð Þ, for i ¼ 1; 2; . . . ;NPDE.

8: RðNPDEÞ – REAL (KIND=nag_wp) array Output

On exit: RðiÞ must be set to the value of Ri x; t; U; Uxð Þ, for i ¼ 1; 2; . . . ;NPDE.

9: IRES – INTEGER Input/Output

On entry: set to �1 or 1.

On exit: should usually remain unchanged. However, you may set IRES to force the
integration routine to take certain actions as described below:

IRES ¼ 2
Indicates to the integrator that control should be passed back immediately to the
calling (sub)routine with the error indicator set to IFAIL ¼ 6.

IRES ¼ 3
Indicates to the integrator that the current time step should be abandoned and a
smaller time step used instead. You may wish to set IRES ¼ 3 when a physically
meaningless input or output value has been generated. If you consecutively set
IRES ¼ 3, then D03PCF/D03PCA returns to the calling subroutine with the error
indicator set to IFAIL ¼ 4.

Note: the following are additional arguments for specific use with D03PCA. Users of
D03PCF therefore need not read the remainder of this description.

10: IUSERð�Þ – INTEGER array User Workspace
11: RUSERð�Þ – REAL (KIND=nag_wp) array User Workspace

PDEDEF is called with the arguments IUSER and RUSER as supplied to D03PCF/
D03PCA. You should use the arrays IUSER and RUSER to supply information to
PDEDEF.

PDEDEF must either be a module subprogram USEd by, or declared as EXTERNAL in, the (sub)
program from which D03PCF/D03PCA is called. Arguments denoted as Input must not be
changed by this procedure.

6: BNDARY – SUBROUTINE, supplied by the user. External Procedure

BNDARY must compute the functions �i and �i which define the boundary conditions as in
equation (3).

The specification of BNDARY for D03PCF is:

SUBROUTINE BNDARY (NPDE, T, U, UX, IBND, BETA, GAMMA, IRES)

INTEGER NPDE, IBND, IRES
REAL (KIND=nag_wp) T, U(NPDE), UX(NPDE), BETA(NPDE),

GAMMA(NPDE)
&

The specification of BNDARY for D03PCA is:

SUBROUTINE BNDARY (NPDE, T, U, UX, IBND, BETA, GAMMA, IRES,
IUSER, RUSER)

&

INTEGER NPDE, IBND, IRES, IUSER(*)
REAL (KIND=nag_wp) T, U(NPDE), UX(NPDE), BETA(NPDE),

GAMMA(NPDE), RUSER(*)
&

1: NPDE – INTEGER Input

On entry: the number of PDEs in the system.

2: T – REAL (KIND=nag_wp) Input

On entry: the current value of the independent variable t.

3: UðNPDEÞ – REAL (KIND=nag_wp) array Input

On entry: UðiÞ contains the value of the component Ui x; tð Þ at the boundary specified
by IBND, for i ¼ 1; 2; . . . ;NPDE.

4: UXðNPDEÞ – REAL (KIND=nag_wp) array Input

On entry: UXðiÞ contains the value of the component
@Ui x; tð Þ

@x
at the boundary

specified by IBND, for i ¼ 1; 2; . . . ;NPDE.

5: IBND – INTEGER Input

On entry: determines the position of the boundary conditions.

IBND ¼ 0
BNDARY must set up the coefficients of the left-hand boundary, x ¼ a.

IBND 6¼ 0
Indicates that BNDARY must set up the coefficients of the right-hand boundary,
x ¼ b.

6: BETAðNPDEÞ – REAL (KIND=nag_wp) array Output

On exit: BETAðiÞ must be set to the value of �i x; tð Þ at the boundary specified by
IBND, for i ¼ 1; 2; . . . ;NPDE.

7: GAMMAðNPDEÞ – REAL (KIND=nag_wp) array Output

On exit: GAMMAðiÞ must be set to the value of �i x; t; U; Uxð Þ at the boundary specified
by IBND, for i ¼ 1; 2; . . . ;NPDE.

8: IRES – INTEGER Input/Output

On entry: set to �1 or 1.

On exit: should usually remain unchanged. However, you may set IRES to force the
integration routine to take certain actions as described below:

IRES ¼ 2
Indicates to the integrator that control should be passed back immediately to the
calling (sub)routine with the error indicator set to IFAIL ¼ 6.

IRES ¼ 3
Indicates to the integrator that the current time step should be abandoned and a
smaller time step used instead. You may wish to set IRES ¼ 3 when a physically
meaningless input or output value has been generated. If you consecutively set
IRES ¼ 3, then D03PCF/D03PCA returns to the calling subroutine with the error
indicator set to IFAIL ¼ 4.

Note: the following are additional arguments for specific use with D03PCA. Users of
D03PCF therefore need not read the remainder of this description.

9: IUSERð�Þ – INTEGER array User Workspace
10: RUSERð�Þ – REAL (KIND=nag_wp) array User Workspace

BNDARY is called with the arguments IUSER and RUSER as supplied to D03PCF/
D03PCA. You should use the arrays IUSER and RUSER to supply information to
BNDARY.

BNDARY must either be a module subprogram USEd by, or declared as EXTERNAL in, the
(sub)program from which D03PCF/D03PCA is called. Arguments denoted as Input must not be
changed by this procedure.

7: UðNPDE;NPTSÞ – REAL (KIND=nag_wp) array Input/Output

On entry: the initial values of U x; tð Þ at t ¼ TS and the mesh points XðjÞ, for j ¼ 1; 2; . . . ;NPTS.

On exit: Uði; jÞ will contain the computed solution at t ¼ TS.

8: NPTS – INTEGER Input

On entry: the number of mesh points in the interval a; b½ �.
Constraint: NPTS � 3.

9: XðNPTSÞ – REAL (KIND=nag_wp) array Input

On entry: the mesh points in the spatial direction. Xð1Þ must specify the left-hand boundary, a,
and XðNPTSÞ must specify the right-hand boundary, b.

Constraint: Xð1Þ < Xð2Þ < � � � < XðNPTSÞ.

10: ACC – REAL (KIND=nag_wp) Input

On entry: a positive quantity for controlling the local error estimate in the time integration. If
E i; jð Þ is the estimated error for Ui at the jth mesh point, the error test is:

E i; jð Þj j ¼ ACC� 1:0þ Uði; jÞj jð Þ:
Constraint: ACC > 0:0.

11: RSAVEðLRSAVEÞ – REAL (KIND=nag_wp) array Communication Array

If IND ¼ 0, RSAVE need not be set on entry.

If IND ¼ 1, RSAVE must be unchanged from the previous call to the routine because it contains
required information about the iteration.

12: LRSAVE – INTEGER Input

On entry: the dimension of the array RSAVE as declared in the (sub)program from which
D03PCF/D03PCA is called.

Constraint: LRSAVE � 6� NPDEþ 10ð Þ � NPDE� NPTSþ 3� NPDEþ 21ð Þ � NPDEþ
7� NPTSþ 54.

13: ISAVEðLISAVEÞ – INTEGER array Communication Array

If IND ¼ 0, ISAVE need not be set on entry.

If IND ¼ 1, ISAVE must be unchanged from the previous call to the routine because it contains
required information about the iteration. In particular:

ISAVEð1Þ
Contains the number of steps taken in time.

ISAVEð2Þ
Contains the number of residual evaluations of the resulting ODE system used. One such
evaluation involves computing the PDE functions at all the mesh points, as well as one
evaluation of the functions in the boundary conditions.

ISAVEð3Þ
Contains the number of Jacobian evaluations performed by the time integrator.

ISAVEð4Þ
Contains the order of the last backward differentiation formula method used.

ISAVEð5Þ
Contains the number of Newton iterations performed by the time integrator. Each iteration
involves an ODE residual evaluation followed by a back-substitution using the LU
decomposition of the Jacobian matrix.

14: LISAVE – INTEGER Input

On entry: the dimension of the array ISAVE as declared in the (sub)program from which
D03PCF/D03PCA is called.

Constraint: LISAVE � NPDE� NPTSþ 24.

15: ITASK – INTEGER Input

On entry: specifies the task to be performed by the ODE integrator.

ITASK ¼ 1
Normal computation of output values U at t ¼ TOUT.

ITASK ¼ 2
One step and return.

ITASK ¼ 3
Stop at first internal integration point at or beyond t ¼ TOUT.

Constraint: ITASK ¼ 1, 2 or 3.

16: ITRACE – INTEGER Input

On entry: the level of trace information required from D03PCF/D03PCA and the underlying
ODE solver. ITRACE may take the value �1, 0, 1, 2 or 3.

ITRACE ¼ �1
No output is generated.

ITRACE ¼ 0
Only warning messages from the PDE solver are printed on the current error message unit
(see X04AAF).

ITRACE > 0
Output from the underlying ODE solver is printed on the current advisory message unit
(see X04ABF). This output contains details of Jacobian entries, the nonlinear iteration and
the time integration during the computation of the ODE system.

If ITRACE < �1, then �1 is assumed and similarly if ITRACE > 3, then 3 is assumed.

The advisory messages are given in greater detail as ITRACE increases. You are advised to set
ITRACE ¼ 0, unless you are experienced with Sub-chapter D02M–N.

17: IND – INTEGER Input/Output

On entry: indicates whether this is a continuation call or a new integration.

IND ¼ 0
Starts or restarts the integration in time.

IND ¼ 1
Continues the integration after an earlier exit from the routine. In this case, only the
arguments TOUT and IFAIL should be reset between calls to D03PCF/D03PCA.

Constraint: IND ¼ 0 or 1.

On exit: IND ¼ 1.

18: IFAIL – INTEGER Input/Output

Note: for D03PCA, IFAIL does not occur in this position in the argument list. See the additional
arguments described below.

On entry: IFAIL must be set to 0, �1 or 1. If you are unfamiliar with this argument you should
refer to Section 3.4 in How to Use the NAG Library and its Documentation for details.

For environments where it might be inappropriate to halt program execution when an error is
detected, the value �1 or 1 is recommended. If the output of error messages is undesirable, then
the value 1 is recommended. Otherwise, if you are not familiar with this argument, the
recommended value is 0. When the value �1 or 1 is used it is essential to test the value of
IFAIL on exit.

On exit: IFAIL ¼ 0 unless the routine detects an error or a warning has been flagged (see
Section 6).

Note: the following are additional arguments for specific use with D03PCA. Users of D03PCF
therefore need not read the remainder of this description.

19: IUSERð�Þ – INTEGER array User Workspace
20: RUSERð�Þ – REAL (KIND=nag_wp) array User Workspace

IUSER and RUSER are not used by D03PCF/D03PCA, but are passed directly to PDEDEF and
BNDARY and should be used to pass information to these routines.

21: CWSAVð10Þ – CHARACTER(80) array Communication Array

If IND ¼ 0, CWSAV need not be set on entry.

If IND ¼ 1, CWSAV must be unchanged from the previous call to the routine.

22: LWSAVð100Þ – LOGICAL array Communication Array

If IND ¼ 0, LWSAV need not be set on entry.

If IND ¼ 1, LWSAV must be unchanged from the previous call to the routine.

23: IWSAVð505Þ – INTEGER array Communication Array

If IND ¼ 0, IWSAV need not be set on entry.

If IND ¼ 1, IWSAV must be unchanged from the previous call to the routine.

24: RWSAVð1100Þ – REAL (KIND=nag_wp) array Communication Array

If IND ¼ 0, RWSAV need not be set on entry.

If IND ¼ 1, RWSAV must be unchanged from the previous call to the routine.

25: IFAIL – INTEGER Input/Output

Note: see the argument description for IFAIL above.

6 Error Indicators and Warnings

If on entry IFAIL ¼ 0 or �1, explanatory error messages are output on the current error message unit
(as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL ¼ 1

On entry, TOUT � TS,
or TOUT� TS is too small,
or ITASK 6¼ 1, 2 or 3,
or M 6¼ 0, 1 or 2,
or M > 0 and Xð1Þ < 0:0,
or the mesh points XðiÞ are not ordered,
or NPTS < 3,
or NPDE < 1,
or ACC � 0:0,
or IND 6¼ 0 or 1,
or LRSAVE is too small,
or LISAVE is too small.

IFAIL ¼ 2

The underlying ODE solver cannot make any further progress across the integration range from
the current point t ¼ TS with the supplied value of ACC. The components of U contain the
computed values at the current point t ¼ TS.

IFAIL ¼ 3

In the underlying ODE solver, there were repeated errors or corrector convergence test failures
on an attempted step, before completing the requested task. The problem may have a singularity
or ACC is too small for the integration to continue. Integration was successful as far as t ¼ TS.

IFAIL ¼ 4

In setting up the ODE system, the internal initialization routine was unable to initialize the
derivative of the ODE system. This could be due to the fact that IRES was repeatedly set to 3 in
at least PDEDEF or BNDARY, when the residual in the underlying ODE solver was being
evaluated.

IFAIL ¼ 5

In solving the ODE system, a singular Jacobian has been encountered. You should check your
problem formulation.

IFAIL ¼ 6

When evaluating the residual in solving the ODE system, IRES was set to 2 in at least PDEDEF
or BNDARY. Integration was successful as far as t ¼ TS.

IFAIL ¼ 7

The value of ACC is so small that the routine is unable to start the integration in time.

IFAIL ¼ 8

In one of PDEDEF or BNDARY, IRES was set to an invalid value.

IFAIL ¼ 9 (D02NNF)

A serious error has occurred in an internal call to the specified routine. Check the problem
specification and all arguments and array dimensions. Setting ITRACE ¼ 1 may provide more
information. If the problem persists, contact NAG.

IFAIL ¼ 10

The required task has been completed, but it is estimated that a small change in ACC is unlikely
to produce any change in the computed solution. (Only applies when you are not operating in one
step mode, that is when ITASK 6¼ 2.)

IFAIL ¼ 11

An error occurred during Jacobian formulation of the ODE system (a more detailed error
description may be directed to the current error message unit).

IFAIL ¼ 12

Not applicable.

IFAIL ¼ 13

Not applicable.

IFAIL ¼ 14

The flux function Ri was detected as depending on time derivatives, which is not permissible.

IFAIL ¼ �99

An unexpected error has been triggered by this routine. Please contact NAG.

See Section 3.9 in How to Use the NAG Library and its Documentation for further information.

IFAIL ¼ �399

Your licence key may have expired or may not have been installed correctly.

See Section 3.8 in How to Use the NAG Library and its Documentation for further information.

IFAIL ¼ �999

Dynamic memory allocation failed.

See Section 3.7 in How to Use the NAG Library and its Documentation for further information.

7 Accuracy

D03PCF/D03PCA controls the accuracy of the integration in the time direction but not the accuracy of
the approximation in space. The spatial accuracy depends on both the number of mesh points and on
their distribution in space. In the time integration only the local error over a single step is controlled
and so the accuracy over a number of steps cannot be guaranteed. You should therefore test the effect of
varying the accuracy argument, ACC.

8 Parallelism and Performance

D03PCF/D03PCA is threaded by NAG for parallel execution in multithreaded implementations of the
NAG Library.

D03PCF/D03PCA makes calls to BLAS and/or LAPACK routines, which may be threaded within the
vendor library used by this implementation. Consult the documentation for the vendor library for further
information.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the
OpenMP environment used within this routine. Please also consult the Users' Note for your
implementation for any additional implementation-specific information.

9 Further Comments

D03PCF/D03PCA is designed to solve parabolic systems (possibly including some elliptic equations)
with second-order derivatives in space. The argument specification allows you to include equations with
only first-order derivatives in the space direction but there is no guarantee that the method of integration
will be satisfactory for such systems. The position and nature of the boundary conditions in particular
are critical in defining a stable problem. It may be advisable in such cases to reduce the whole system
to first-order and to use the Keller box scheme routine D03PEF.

The time taken depends on the complexity of the parabolic system and on the accuracy requested.

10 Example

We use the example given in Dew and Walsh (1981) which consists of an elliptic-parabolic pair of
PDEs. The problem was originally derived from a single third-order in space PDE. The elliptic equation
is

1

r

@

@r
r2
@U1

@r

� �
¼ 4� U2 þ r

@U2

@r

� �

and the parabolic equation is

1� r2
� �@U2

@t
¼ 1

r

@

@r
r

@U2

@r
� U2U1

� �� �

where r; tð Þ 2 0; 1½ � � 0; 1½ �. The boundary conditions are given by

U1 ¼ @U2

@r
¼ 0 at r ¼ 0;

and

@

@r
rU1ð Þ ¼ 0 and U2 ¼ 0 at r ¼ 1:

The first of these boundary conditions implies that the flux term in the second PDE,
@U2

@r
� U2U1

� �
, is

zero at r ¼ 0.

The initial conditions at t ¼ 0 are given by

U1 ¼ 2�r and U2 ¼ 1:0; r 2 0; 1½ �:
The value � ¼ 1 was used in the problem definition. A mesh of 20 points was used with a circular mesh
spacing to cluster the points towards the right-hand side of the spatial interval, r ¼ 1.

10.1 Program Text

the following program illustrates the use of D03PCF. An equivalent program illustrating the use of
D03PCA is available with the supplied Library and is also available from the NAG web site.

! D03PCF Example Program Text
! Mark 26 Release. NAG Copyright 2016.

Module d03pcfe_mod

! D03PCF Example Program Module:
! Parameters and User-defined Routines

! .. Use Statements ..
Use nag_library, Only: nag_wp

! .. Implicit None Statement ..
Implicit None

! .. Accessibility Statements ..
Private
Public :: bndary, pdedef, uinit

! .. Parameters ..

Integer, Parameter, Public :: nin = 5, nout = 6, npde = 2
! .. Local Scalars ..

Real (Kind=nag_wp), Public, Save :: alpha
Contains

Subroutine pdedef(npde,t,x,u,ux,p,q,r,ires)

! .. Scalar Arguments ..
Real (Kind=nag_wp), Intent (In) :: t, x
Integer, Intent (Inout) :: ires
Integer, Intent (In) :: npde

! .. Array Arguments ..
Real (Kind=nag_wp), Intent (Out) :: p(npde,npde), q(npde), r(npde)
Real (Kind=nag_wp), Intent (In) :: u(npde), ux(npde)

! .. Executable Statements ..
q(1) = 4.0_nag_wp*alpha*(u(2)+x*ux(2))
q(2) = 0.0_nag_wp
r(1) = x*ux(1)
r(2) = ux(2) - u(1)*u(2)
p(1,1) = 0.0_nag_wp
p(1,2) = 0.0_nag_wp
p(2,1) = 0.0_nag_wp
p(2,2) = 1.0_nag_wp - x*x
Return

End Subroutine pdedef
Subroutine bndary(npde,t,u,ux,ibnd,beta,gamma,ires)

! .. Scalar Arguments ..
Real (Kind=nag_wp), Intent (In) :: t
Integer, Intent (In) :: ibnd, npde
Integer, Intent (Inout) :: ires

! .. Array Arguments ..
Real (Kind=nag_wp), Intent (Out) :: beta(npde), gamma(npde)
Real (Kind=nag_wp), Intent (In) :: u(npde), ux(npde)

! .. Executable Statements ..
If (ibnd==0) Then

beta(1) = 0.0_nag_wp
beta(2) = 1.0_nag_wp
gamma(1) = u(1)
gamma(2) = -u(1)*u(2)

Else
beta(1) = 1.0_nag_wp
beta(2) = 0.0_nag_wp
gamma(1) = -u(1)
gamma(2) = u(2)

End If
Return

End Subroutine bndary
Subroutine uinit(u,x,npts)

! Routine for PDE initial condition

! .. Scalar Arguments ..
Integer, Intent (In) :: npts

! .. Array Arguments ..
Real (Kind=nag_wp), Intent (Out) :: u(2,npts)
Real (Kind=nag_wp), Intent (In) :: x(npts)

! .. Local Scalars ..
Integer :: i

! .. Executable Statements ..
Do i = 1, npts

u(1,i) = 2.0_nag_wp*alpha*x(i)
u(2,i) = 1.0_nag_wp

End Do
Return

End Subroutine uinit
End Module d03pcfe_mod
Program d03pcfe

! D03PCF Example Main Program

! .. Use Statements ..

Use nag_library, Only: d03pcf, d03pzf, nag_wp, x01aaf
Use d03pcfe_mod, Only: alpha, bndary, nin, nout, npde, pdedef, uinit

! .. Implicit None Statement ..
Implicit None

! .. Local Scalars ..
Real (Kind=nag_wp) :: acc, hx, pi, piby2, tout, ts
Integer :: i, ifail, ind, intpts, it, itask, &

itrace, itype, lisave, lrsave, m, &
neqn, npts, nwk

! .. Local Arrays ..
Real (Kind=nag_wp), Allocatable :: rsave(:), u(:,:), uout(:,:,:), x(:), &

xout(:)
Integer, Allocatable :: isave(:)

! .. Intrinsic Procedures ..
Intrinsic :: real, sin

! .. Executable Statements ..
Write (nout,*) ’D03PCF Example Program Results’

! Skip heading in data file
Read (nin,*)
Read (nin,*) intpts, npts, itype
neqn = npde*npts
lisave = neqn + 24
nwk = (10+6*npde)*neqn
lrsave = nwk + (21+3*npde)*npde + 7*npts + 54
Allocate (rsave(lrsave),u(npde,npts),uout(npde,intpts,itype),x(npts), &

xout(intpts),isave(lisave))

Read (nin,*) xout(1:intpts)
Read (nin,*) acc, alpha
Read (nin,*) m, itrace
ind = 0
itask = 1

! Set spatial mesh points

piby2 = 0.5_nag_wp*x01aaf(pi)
hx = piby2/real(npts-1,kind=nag_wp)
x(1) = 0.0_nag_wp
x(npts) = 1.0_nag_wp
Do i = 2, npts - 1

x(i) = sin(hx*real(i-1,kind=nag_wp))
End Do

! Set initial conditions
Read (nin,*) ts, tout

! Set the initial values
Call uinit(u,x,npts)

Do it = 1, 5
tout = 10.0_nag_wp*tout

! ifail: behaviour on error exit
! =0 for hard exit, =1 for quiet-soft, =-1 for noisy-soft

ifail = 0
Call d03pcf(npde,m,ts,tout,pdedef,bndary,u,npts,x,acc,rsave,lrsave, &

isave,lisave,itask,itrace,ind,ifail)

If (it==1) Then
Write (nout,99999) acc, alpha
Write (nout,99998) xout(1:6)

End If

! Interpolate at required spatial points

ifail = 0
Call d03pzf(npde,m,u,npts,x,xout,intpts,itype,uout,ifail)

Write (nout,99996) tout, uout(1,1:intpts,1)
Write (nout,99995) uout(2,1:intpts,1)

End Do

! Print integration statistics

Write (nout,99997) isave(1), isave(2), isave(3), isave(5)

99999 Format (/,/,’ Accuracy requirement = ’,E12.5,/,’ Parameter ALPHA =’, &
’ ’,E12.3,/)

99998 Format (’ T / X ’,6F8.4,/)
99997 Format (’ Number of integration steps in time ’,I4,/, &

’ Number of residual evaluations of resulting ODE system’,I4,/, &
’ Number of Jacobian evaluations ’,I4,/, &
’ Number of iterations of nonlinear solver ’,I4)

99996 Format (1X,F7.4,’ U(1)’,6F8.4)
99995 Format (9X,’U(2)’,6F8.4,/)

End Program d03pcfe

10.2 Program Data

D03PCF Example Program Data
6 20 1 : intpts, npts, itype
0.0 0.4 0.6 0.8 0.9 1.0 : xout(1:intpts)
1.0E-3 1.0 : acc, alpha
1 0 : m, itrace
0.0 0.1E-4 : ts, tout

10.3 Program Results

D03PCF Example Program Results

Accuracy requirement = 0.10000E-02
Parameter ALPHA = 0.100E+01

T / X 0.0000 0.4000 0.6000 0.8000 0.9000 1.0000

0.0001 U(1) 0.0000 0.8008 1.1988 1.5990 1.7958 1.8485
U(2) 0.9997 0.9995 0.9994 0.9988 0.9663 -0.0000

0.0010 U(1) 0.0000 0.7982 1.1940 1.5841 1.7179 1.6734
U(2) 0.9969 0.9952 0.9937 0.9484 0.6385 -0.0000

0.0100 U(1) 0.0000 0.7676 1.1239 1.3547 1.3635 1.2830
U(2) 0.9627 0.9495 0.8754 0.5537 0.2908 -0.0000

0.1000 U(1) 0.0000 0.3908 0.5007 0.5297 0.5120 0.4744
U(2) 0.5468 0.4299 0.2995 0.1479 0.0724 -0.0000

1.0000 U(1) 0.0000 0.0007 0.0008 0.0008 0.0008 0.0007
U(2) 0.0010 0.0007 0.0005 0.0002 0.0001 -0.0000

Number of integration steps in time 78
Number of residual evaluations of resulting ODE system 378
Number of Jacobian evaluations 25
Number of iterations of nonlinear solver 190

Example Program
Solution, U(1,x,t), of Elliptic-parabolic Pair using Method of Lines and BDF Method

Solution, U(2,x,t), of Elliptic-parabolic Pair using Finite-differences and BDF

